



SOUNDING BOARD

Jenny Schmidt

Generating a wide variety of sounds from your Apple II can be very frustrating. In Applesoft you're limited to the bell character (Control-G) or some very dull speaker-clicking via the PEEK(-16336) command. Assembly language broadens your possibilities, but you have to know what you're doing. Even then, to obtain predictable results takes patience, time, and a lot of bit fiddling.

The "Sounding Board" program makes things easy by giving you complete control over a sound's pitch, duration, and envelope (volume of a sound through time). The program generates *sound tables* which may be heard using Sounding Board's Play option or played in your own programs through use of "VTone"—the accompanying machine language routine that adds sound commands to BASIC.

Probably the best thing about this program is that you design your sounds visually. Through use of the keyboard, you stretch and shrink the sound's envelope by stretching and shrinking a line that's displayed on the computer screen. Options such as setting sustain and noise levels allow you to shape the sound even further. With some thought and a little experimentation, you can design sounds that mimic musical instruments or just make unusual noises.

The Sound of Typing

To use Sounding Board, you need to enter and save Programs 1 and 2. Program 1 is the sound editor. It is written in BASIC. Accurate typing is important here, so be sure to use the "Apple Automatic Proofreader," found elsewhere in this issue, when typing in Program 1. Save this program as **SOUND.EDITOR**.

Program 2, "VTone," is the heart of Sounding Board, providing all the key machine language routines that are used to produce sound. You must enter this listing using "Apple MLX," the machine language entry program listed elsewhere in this magazine. Before loading Apple MLX, type

HIMEM: 35840

When you run Apple MLX, you'll be asked for the starting and ending addresses of the program that you're about to enter. For Program 2, answer these prompts with

STARTING ADDRESS? 9400

ENDING ADDRESS? 953F

After Apple MLX displays the options menu, choose E to enter the program and then type in your starting address. (If you're just beginning to enter VTone, type **9400**, the first ad-

dress in the listing.) Enter the data and save the file with the filename **VTONE** to the same disk as Program 1.

Note: It's very important that you save this file as VTONE; Program 1 expects to find Program 2 saved with that filename.

Making Noise

You don't need to know how the Apple II produces sound to use Sounding Board, but knowing how the program generates different tones will make it easier for you to create the sounds you want.

Sounding Board's sound synthesis process relies on the use of sound tables. A sound table is a list of numeric data representing the *volume envelope* for a sound. A volume envelope is simply the loudness of a sound over time. For example, consider how a piano note is created: The player presses a key and a hammer strikes a stretched string. The hammer imparts energy to the string, and the string starts vibrating. These vibrations are the sound you hear. The sound begins loud, but over time the vibrations become weaker and weaker, and the sound becomes softer and softer until it's no longer heard.

If you drew a graph of the volume level of a piano note over time, it would look somewhat like the figure shown below. This graph is a pictorial representation of the volume envelope for a piano note. The volume envelope is a very important component in determining the characteristics of a sound.

With Sounding Board, you actually *draw* the sound's volume envelope on the computer screen. When you're done, the program calculates a sound table for this envelope. Then you can "play" the sound table, hear what it sounds like, and if you like, go back and edit it until you get it just right.

Becoming an Audio Artist

To begin designing sounds, enter **RUN SOUND.EDITOR**. After a short while, the program's main screen appears (see screen shot). The line in the middle of the screen illustrates the current envelope. Using the I, J, K, and M keys, you can change this line into a picture of the sound you want.

The computer screen acts as a graph of the sound's envelope, with the volume recorded vertically and the time recorded horizontally. The higher the line is, the louder the sound's volume. Time runs left to right across the screen. Near the bottom of the screen, above the menu, is a series of numbers. These numbers mark time in milliseconds.

**Get machine-language-quality sounds from BASIC
with "Sounding Board"—a sound design program
which lets you draw sounds onscreen. Works on the
Apple IIe, IIc, or IIGSGS in either DOS 3.3 or ProDOS,
and on the Apple II+ with DOS 3.3.**



The line is actually made up of 32 connecting segments. Think of the line as a long, hinged ruler that you can push and bend into shape. A square on the screen indicates your current position on the line. The J and K keys move this square cursor left and right along the line. When you come to the end of the line, the cursor wraps around to the other side of the screen. The I and M keys pull the line up and down at the cursor's current position.

The Menu


Sounding Board's menu appears below the graphing area. Options include:

NEW	SUST	TIME	PLAY	NOISE
LOAD	SAVE	CAT	DRIVE	QUIT

You select options by pressing the left and right arrow keys. Brackets outline the current option. Press Return to activate the selected menu option.

NEW. This option erases the current envelope and returns it to a straight line.

SUST. Although you can create many different sounds by changing the envelope, there are other options that affect the sound: SUST is one of them.



The main screen of "Sounding Board" shows the sound envelope above and lists the menu options below.


The SUST option allows you to indicate a *sustain level* in the envelope. The sustain level is a constant volume that lasts for nearly all of the sound's duration. For example, consider an oboe. When an oboe plays a note, the volume builds quickly and then stays at its sustain level. Finally, the player stops and the volume drops off quickly. The length of time that the volume is sustained varies with the duration of the note.

When you select SUST, a caret symbol (^) appears at the bottom of the envelope graph, just above the time scale. Each time you select SUST, the caret moves one space to the right. Keep moving it until it reaches the point in time where you want the sustain to occur. Now when you play a note, the sustain occurs at the time you indicated and at the current volume of the envelope.

The sustain lasts for the remainder of the note's normal duration, and then the rest of envelope is played. As you can see, setting a sustain causes the note to be slightly longer than the specified duration. The length of the note is actually the duration plus the remainder of the envelope. Since sustains are usually used with very short envelopes, the extra length is seldom noticed.

To remove a sustain, continue to select SUST until the caret moves off the screen.

VOLUME ENVELOPE FOR PIANO NOTE



TIME. If you look at the piano envelope in the figure shown above, you'll notice that it spans 600 milliseconds. When you first run SOUND.EDITOR, the envelope's time span defaults to 100 milliseconds. To increase or decrease the envelope time, select this option. You may choose any time between 40 and 2100 milliseconds. Generally, shorter time scales make for more percussive sounds. The time scale at the bottom of the graph reflects your changes.

PLAY. Once you've drawn an envelope using the I, J, K, and M keys, you'll want to hear how it sounds. With this option, you can do just that.

When you select PLAY for the first time, you must wait while the program calculates the envelope's new sound table. Then you're asked to enter a sequence of notes. Each note is composed of a pitch value and a duration value. These values can range between 1 and 255. The smaller the pitch value, the higher the pitch. The larger the duration value, the longer the note's duration.

Every note must be entered onto one line. To separate the pitch from the duration and to separate each note, you must use an exclamation point (!). So your entry for two different notes would appear as pitch!duration!pitch!duration. For example, the following sequence plays the popular *do-re-mi* arpeggio (otherwise known as the first three notes in a C-major scale: C, D, and E):

ENTER SEQUENCE: 76!100!63!100!52!100

Each note here is of equal duration (100). The Note Values Table lists the pitch values for notes A2-F5 (A in the second octave through F in the fifth octave), where C4 represents middle C. A pitch value of 0 indicates a rest (no

NOTE VALUES

Second Octave		Third Octave		Fourth Octave		Fifth Octave	
Note	Value	Note	Value	Note	Value	Note	Value
C	188	C	76	C	17		
C#/Db	181	C#/Db	69	C#/Db	14		
D	169	D	63	D	11		
D#/Eb	155	D#/Eb	57	D#/Eb	8		
E	145	E	52	E	5		
F	135	F	47	F	2		
F#/Gb	125	F#/Gb	42				
G	114	G	36				
G#/Ab	107	G#/Ab	32				
A	239	A	98	A	28		
A#/Bb	223	A#/Bb	91	A#/Bb	24		
B	208	B	83	B	21		

sound). A note's duration can be calculated by multiplying the duration value by 8.278 milliseconds. A duration of 120, for example, is about one second long.

You can enter a new sequence after the current notes have been played. Press RETURN without entering a sequence to return to the menu.

NOISE. NOISE creates unusual sound effects by randomly scrambling a sound's pitch. To hear how it works, select NOISE and enter a value 0-255; then choose PLAY. A NOISE value of 0 indicates no noise, while 255 selects the greatest amount of noise. If you enter 255, all the notes you play will sound like static. If you choose an intermediate value, you'll hear more pitch than static in the note. Experiment with different values until you get a feel for how the NOISE option affects the sound.

LOAD. This option loads sounds created by Sounding Board so you can edit them further.

SAVE. When you design a sound, you can save it to disk for later editing or use in your own programs. Two files are created when you save a sound to disk: one using the filename you entered and the other with a .S extension.

CAT. To get a catalog of the current drive, select this option.

DRIVE. When the program first runs, it defaults to using drive 1, slot 6. Using this option, you can switch between drive 1 and drive 2.

QUIT. Choose this to exit the program.

Sounding Off with BASIC

So far you've been playing sounds from inside the sound editor. Eventually, you'll want to use these sounds in your own programs. To do this, you must save your sounds to disk using the editor's SAVE option.

To access sounds from BASIC, first load the machine language sound routine VTone into memory, along with any sound files that you wish to use. The following BASIC code accomplishes just that, loading the example sound files PIANO and SQUISH:

```
10 PRINT CHR$(4); "BLOAD VTONE"
20 PRINT CHR$(4); "BLOAD PIANO,A37632"
30 PRINT CHR$(4); "BLOAD SQUISH,A37376"
```

For your own use, you should substitute for PIANO and SQUISH the filenames of the sound files that you wish to use.

You can have as many sound tables in memory as you want. Just BLOAD them in, one below the other in memory. Each sound table takes up 256 bytes of memory. Because VTone occupies memory locations 37888-38206, the first sound file—PIANO in the example above—should be loaded into memory at 37632 (37888 - 256 = 37632). After loading in all the sounds that you want, you must set HIMEM to protect VTone and the sound tables from being overwritten by BASIC.

40 HIMEM:37376

Here, HIMEM is set to 37376 because the last sound table was loaded at this location.

Before playing a sound, you should specify what noise level you want the sound to have. (This information is not included in the sound's disk file.) To set the noise level, use the command CALL 38109,noise. The following line sets the noise level to 0 (no noise):

50 CALL 38109,0

Before you can play a sound, you must tell VTone where the sound table is located in memory. For this, use the command CALL 38117,address. In line 20, we load PIANO at 37632. So, to tell VTone where PIANO is, use the following line:

60 CALL 38117,37632

To play notes using the current sound, enter the command CALL 38083,sound parameters. The sound parameters consist of a pitch-duration value for each note—just like the sound editor's play option. Except here, you must use commas to separate each item, instead of exclamation points. For example, the following line plays *do, re, mi* using a duration of 100:

```
70 CALL 38083,76,100,63,100,52,100
```

To play the same notes using the SQUISH sound and a noise level of 15, use these lines:

```
80 CALL 38109,15
```

```
90 CALL 38117,37376
```

```
100 CALL 38083,76,100,63,100,52,100
```

How It Works

You can use Sounding Board's VTone routine without knowing how it works, but you may be interested in how VTone controls the volume of the speaker. After all, Apple manuals rarely mention anything about sound volume.

When you toggle the Apple's speaker through machine language, the speaker responds by moving outward toward a maximum *out* position. When toggled again, the speaker returns to its *resting* position. If the second toggle occurs before the speaker reaches its maximum *out* position, the click is not as loud. Thus, volume can be controlled by carefully timing the second speaker toggle.

VTone uses data from the sound tables to figure out how much time to wait before toggling the speaker for the second time. This way, each sound can have a unique volume level for any point in time, creating a wide variety of beeps, boinks, and buzzes.

Program Key

Key	Function
I	Move envelope up
M	Move envelope down
J	Move left along envelope
K	Move right along envelope
Left Arrow	Select menu option
Right Arrow	Select menu option
Return	Activate menu option

Sounding Board—Sound Editor

Be sure to use "Apple Automatic Proofreader," found elsewhere in this issue, to enter the following program.

```
7F 100 REM COPYRIGHT 1988 COMPUTE! PUBLICATIONS, INC. - ALL RIGHTS RESERVED
EC 110 IF PEEK (104) < 64 THEN POKE 104,64
: POKE 16384,0: PRINT CHR$ (4) "RUN SOUND.EDITOR"
88 120 HOME : TEXT : VTAB 10: HTAB 12: INVERSE : PRINT " SOUNDING BOARD ": NO RMAL
D7 130 VTAB 12: HTAB 13: PRINT "COPYRIGHT 1988": HTAB 7: PRINT "COMPUTE! PUBLICATIONS, INC.": HTAB 10: PRINT "ALL RIGHTS RESERVED"
2F 140 ONERR GOTO 1100
62 150 PRINT CHR$ (4); "BLOAD VTONE"
C9 160 HIMEM: 24576
1B 170 BL = 1: DD = 1
06 180 DEF FN VO(I) = VL(31 * I / NU + 1)
* (1 - I * 31 / NU + INT (I * 31 / NU)) + VL(31 * I / NU + 2) * (I * 31 / NU - INT (I * 31 / NU))
```

```

BA 190 PLAY = 38083: NS = 38109: TB = 38117:
SV = 5: TI = 100: MEM = 30720: IN = 1:
JX = 0: JY = 80: DIM VL(32): SC = 1.5
875: FOR W = 1 TO 32: VL(W) = 80: NE
XT W
25 200 REM POKE SHAPE TABLE AND ERROR ROUT
INE
00 210 DATA 1,0,4,0,33,63,54,45,4,0,104,16
8,104,166,223,154,72,152,72,96
79 220 FOR W = 0 TO 19: READ I: POKE 768 +
W, I: NEXT W: POKE 232,0: POKE 233,
3: ROT= 0: SCALE= 2
E5 230 CALL TB,30720: CALL NS,0
9F 240 HGR : HCOLOR= 3: HPLOT 0,80 TO 279,
80: XDRAW 1 AT JX,JY
56 250 REM MAIN MENU
D6 260 BX = 1: BY = 23: FL = 1
A0 270 HOME : VTAB 23: HTAB 2: PRINT "NEW"
;: HTAB 10: PRINT "SUST";: HTAB 18:
PRINT "TIME";: HTAB 26: PRINT "PLA
Y";: HTAB 34: PRINT "NOISE"
ID 280 HTAB 2: PRINT "LOAD";: HTAB 10: PRI
NT "SAVE";: HTAB 18: PRINT "CAT";:
HTAB 26: PRINT "DRIVE";: HTAB 34: P
RINT "QUIT";
77 290 IF ST < > 0 THEN VTAB 21: HTAB ST:
PRINT "^";
6E 300 VTAB 22: HTAB 1: PRINT "0";: HTAB 1
2: PRINT INT (TI / 40 * 12);: HTAB
24: PRINT INT (TI / 40 * 24);: HTAB
36: PRINT INT (TI / 40 * 36)
10 310 POKE -16368,0
BB 320 HTAB BX: VTAB BY: PRINT "<";: HTAB
BX + 6: PRINT ">";
18 330 IF PEEK (-16384) < 128 THEN 330
5B 340 KY = PEEK (-16384)
18 350 POKE -16368,0
D8 360 IF KY < > 149 THEN 390
4C 370 HTAB BX: VTAB BY: PRINT " ";: HTAB
BX + 6: VTAB BY: PRINT " ";: BX = BX
+ 8: IF BX > 33 THEN BX = 1: BY = B
Y + 1: IF BY > 24 THEN BY = 23
9F 380 GOTO 320
B7 390 IF KY < > 136 THEN 430
8F 400 HTAB BX: VTAB BY: PRINT " ";: HTAB
BX + 6: VTAB BY: PRINT " ";: BX = BX
- 8: IF BX < 1 THEN :BY = BY - 1: B
X = 33: IF BY < 23 THEN BY = 24
92 410 GOTO 320
6A 420 REM CHOOSE OPTION
4C 430 IF KY = 141 THEN ON (BY - 23) * 5 +
(BX - 1) / 8 + 1 GOTO 600,630,680,
730,1060,770,860,970,1030,940
1E 440 REM CHOOSE CURSOR INCREMENT
CE 450 IF KY > 176 AND KY < 186 THEN SV =
KY - 176: GOTO 320
A8 460 REM PROCESS I,J,K,M
85 470 IF KY < > 202 THEN 500
35 480 XDRAW 1 AT JX,JY: IN = IN - 1: JX = J
X - 9: IF ( NOT IN) THEN IN = 32: JX =
279
73 490 JY = VL(IN): XDRAW 1 AT JX,JY: GOTO
320
86 500 IF KY < > 203 THEN 530
AC 510 XDRAW 1 AT JX,JY: IN = IN + 1: JX = J
X + 9: IF IN = 33 THEN IN = 1: JX =
0
10 520 GOTO 490
A2 530 IF KY < > 205 THEN 560
E4 540 XDRAW 1 AT JX,JY: HCOLOR= 0: GOSUB
1380: VL(IN) = VL(IN) + SV: IF VL(IN)
> 159 THEN VL(IN) = 159
BE 550 FL = 1: JY = VL(IN): HCOLOR= 3: GOSU
B 1380: XDRAW 1 AT JX,JY: GOTO 320
7E 560 IF KY < > 201 THEN 320
A7 570 XDRAW 1 AT JX,JY: HCOLOR= 0: GOSUB
1380: VL(IN) = VL(IN) - SV: IF VL(IN)
> 0 THEN VL(IN) = 0
A5 580 GOTO 550
DA 590 REM NEW OPTION
7F 600 HOME : VTAB 22: INPUT "ERASE ENVELO
PE? "; A$: IF LEFT$ (A$,1) < > "Y" A
ND LEFT$ (A$,1) < > CHR$ (121) THEN
270
77 610 FOR W = 1 TO 32: VL(W) = 80: NEXT W:
JX = 0: JY = 80: IN = 1: GOTO 240
68 620 REM SUSTAIN OPTION
25 630 IF ST THEN HTAB ST: VTAB 21: PRINT
" ";
8E 640 ST = ST + 1: IF ST > 40 THEN ST = 0
A5 650 IF ST THEN HTAB ST: VTAB 21: PRINT
"~";
62 660 FL = 1: GOTO 320
80 670 REM TIME OPTION
93 680 HOME : VTAB 22: PRINT "CURRENT TIME
IS "; TI; " MILLISECONDS": INPUT "EN
TER TIME IN MILLISECONDS (40-2100)
": A$: IF A$ = "" THEN 270
5A 690 IF VAL (A$) < 40 OR VAL (A$) > 2100
THEN 680
8A 700 TI = VAL (A$): FL = 1: GOTO 270
95 710 GOTO 320
44 720 REM PLAY OPTION
A6 730 GOSUB 1300: HOME : VTAB 21
8E 740 PRINT "ENTER SEQUENCE": INPUT ":"; A
$: IF A$ = "" THEN 270
EF 750 A$ = "!" + A$ + CHR$ (0): CALL 3816
3, A$: GOTO 740
A2 760 REM LOAD OPTION
31 770 TEXT : HOME : VTAB 12: INPUT "ENTER
FILE NAME (?=CATALOG, <RET> TO
CANCEL) : "; A$: PRINT
F3 780 IF A$ = "" THEN POKE -16304,0: GOT
O 270
8A 790 IF A$ = "?" OR A$ = "/" THEN GOSUB
980: GOTO 770
5E 800 INPUT "LOADING WILL ERASE CURRENT E
NVELOPE. DO YOU WANT TO CONTINUE? "
; B$: IF LEFT$ (B$,1) = "N" OR LEFT$ (B$,1) =
CHR$ (ASC ("N") + 32) TH
EN POKE -16304,0: GOTO 270
8C 810 PRINT CHR$ (4)"VERIFY"; A$; ".S,D"; DD
F4 820 PRINT CHR$ (4); "OPEN"; A$; ".S,D"; DD:
PRINT CHR$ (4); "READ"; A$; ".S": FOR
I = 1 TO 32: INPUT VL(I): NEXT I:
INPUT ST: INPUT TI: INPUT NI: PRINT
CHR$ (4); "CLOSE"
F8 830 HGR : HCOLOR= 3: FOR I = 0 TO 30: H
PLOT I * 9, VL(I + 1) TO I * 9 + 9, V
L(I + 2): NEXT I
5E 840 JX = 0: JY = VL(1): XDRAW 1 AT JX,JY
: FL = 1: GOTO 270
73 850 REM SAVE OPTION
67 860 TEXT : HOME
32 870 VTAB 12: INPUT "ENTER FILE NAME (?=
CATALOG, <RET> TO CANCEL) : "; A
$: PRINT
F4 880 IF A$ = "" THEN POKE -16304,0: GOT
O 270
7B 890 IF A$ = "?" OR A$ = "/" THEN GOSUB
980: GOTO 860
CB 900 GOSUB 1300
25 910 PRINT CHR$ (4); "OPEN"; A$; ".S,D"; DD:
PRINT CHR$ (4); "WRITE"; A$; ".S": FO
R I = 1 TO 32: PRINT VL(I): NEXT I:
PRINT ST: PRINT TI: PRINT NI
27 920 PRINT CHR$ (4); "CLOSE": PRINT CHR$ (4);
"BSAVE "; A$; ",A30720,L$100 ,D";

```

```

DD: POKE - 16304,0: GOTO 270
8B 930 REM QUIT OPTION
6C 940 HOME : VTAB 22: INPUT "EXIT PROGRAM
? ";A$: IF LEFT$ (A$,1) < > "Y" AND
LEFT$ (A$,1) < > CHR$ (121) THEN 2
70
E0 950 TEXT : HOME : END
2C 960 REM CATALOG OPTION
B4 970 GOSUB 980: GOTO 270
6C 980 TEXT : HOME
AA 990 IF PEEK (48896) = 76 THEN PRINT CHR
$ (4)"CAT": GOTO 1010
17 1000 PRINT : PRINT CHR$ (4)"CATALOG"
8A 1010 GET A$: POKE - 16304,0: RETURN
AD 1020 REM DRIVE OPTION
11 1030 TEXT : HOME : VTAB 12: INPUT "ENTE
R DRIVE NUMBER (1 OR 2) : ";A$:DD
= VAL (A$): IF DD < > 1 AND DD < >
2 THEN 1030
C0 1040 POKE - 16304,0: GOTO 270
AF 1050 REM NOISE OPTION
BD 1060 HOME : VTAB 22: PRINT "CURRENT NOI
SE VALUE IS ";NI: INPUT "ENTER NOI
SE VALUE (0-255)";A$: IF A$ = ""
THEN 270
AA 1070 NI = VAL (A$): IF NI < 0 OR NI > 2
55 THEN 1060
FE 1080 CALL NS,NI: GOTO 270
88 1090 REM ERROR HANDLER
35 1100 IF BL THEN 1160
E2 1110 PRINT "VTONE FILE NOT ON DISK. INS
ERT CORRECT DISK AND PRESS <RET>.
PRESS <ESC> TO QUIT.": POKE -
16368,0
5C 1120 KY = PEEK (- 16384)
D8 1130 IF KY = 141 THEN POKE - 16368,0: G
OTO 150
E8 1140 IF KY = 155 THEN POKE - 16368,0: E
ND
6A 1150 GOTO 1120
8E 1160 IF PEEK (218) + PEEK (219) * 256 =
630 THEN CALL 778: GOTO 740: REM
IF ERR IN LINE 630 THEN GOTO 620
A9 1170 HOME : VTAB 12: ON PEEK (222) GOTO
1180,1180,1180,1190,1200,1210,118
0,1220,1230,1240,1250,1180,1180,11
80,1180,1260
15 1180 PRINT "ERROR #"; PEEK (222): END
B1 1190 PRINT "DISK IS WRITE PROTECTED": G
OTO 1270
78 1200 PRINT "FILE IS NOT ON DISK": PRINT
CHR$ (4); "DELETE";A$: GOTO 1270
AF 1210 PRINT "FILE IS NOT ON DISK": GOTO
1270
20 1220 PRINT "I/O ERROR": GOTO 1270
51 1230 PRINT "DISK FULL": GOTO 1270
DB 1240 PRINT "FILE IS LOCKED": GOTO 1270
15 1250 PRINT "INVALID FILE NAME": GOTO 12
70
19 1260 PRINT "INVALID FILE NAME": GOTO 12
70
48 1270 INPUT "PRESS RETURN TO CONTINUE";A
$
92 1280 POKE - 16304,0: CALL 778: GOTO 270
8B 1290 REM CREATE ENVELOPE ROUTINE
5B 1300 IF NOT FL THEN RETURN
FA 1310 HOME : VTAB 21
2E 1320 PRINT "PLEASE WAIT"
46 1330 NU = INT (TI / 8.2784223) - 1: FOR
I = 0 TO NU - 1: POKE MEM + I, 254
- FN VO(I) * SC: NEXT I
EC 1340 POKE MEM + I, 255: POKE MEM + I + 1
, 255: DR = INT (NU * ST / 40 - 1):
IF DR < 1 THEN DR = 1

```

```

33 1350 IF ST THEN POKE MEM + DR,0
D0 1360 FL = 0: RETURN
ED 1370 REM MOVE LINE SEGMENTS ROUTINE
C1 1380 IF IN > 1 THEN HPLT JX - 9,VL(IN
- 1) TO JX,JY
56 1390 IF IN < 32 THEN HPLT JX,JY TO JX
+ 9,VL(IN + 1)
D9 1400 RETURN

```

Sounding Board—VTone

For mistake-proof entry, use "Apple MLX," found elsewhere in this issue, to type in this program.

```

9400: 86 FD A9 00 85 FC 85 FA 47
9408: 85 FB A4 FB B1 07 F0 60 33
9410: C9 FF F0 5B 85 FE 85 FF 25
9418: EA 20 5F 94 A5 FA 18 69 A6
9420: 2A 85 FA A5 FB 69 00 85 84
9428: FB C5 FD F0 42 48 68 EA 7E
9430: EA EA EA A5 CE 0A B0 C2
9438: 04 49 1D 90 02 EA EA 85 79
9440: CE 25 09 45 06 AA A5 FA B0
9448: 18 69 01 85 FA A5 FB 69 20
9450: 00 85 FB C5 FD F0 18 85 20
9458: FF EA CA D0 E9 F0 AB A8 B5
9460: 49 FF AA CA D0 FD AD 30 3A
9468: C0 88 D0 FD AD 30 C0 60 1E
9470: E6 FB 4C 7C 94 48 68 48 3C
9478: 68 EA 85 FF A5 FE 20 5F 0A
9480: 94 A5 FC 38 E9 2A 85 FC 80
9488: A5 FD E9 00 85 FD 20 6F 15
9490: 94 85 FF F0 9D EA EA EA CD
9498: A5 CE 0A B0 04 49 1D 90 A4
94A0: 02 EA EA 85 CE 25 09 45 9D
94A8: 06 AA A5 FC 38 E9 01 85 F4
94B0: FC A5 FD E9 00 85 FD 85 B7
94B8: FF EA F0 9D 85 FF CA D0 27
94C0: E9 F0 B2 20 F5 E6 86 06 D1
94C8: 8A F0 2E 20 F5 E6 20 00 C6
94D0: 94 A6 FD F0 02 D0 25 20 7A
94D8: B7 00 D0 E7 60 20 4C E7 7A
94E0: 86 09 4C B7 00 20 BE DE 71
94E8: 20 7B DD 20 52 E7 A5 50 8C
94F0: 85 07 A5 51 85 08 4C B7 05
94F8: 00 20 F5 E6 A9 32 20 A8 56
9500: FC A9 06 20 A8 FC 20 6F BF
9508: 94 48 68 EA EA EA CA D0 B4
9510: EB F0 C4 20 BE DE 20 E3 9D
9518: DF A5 B8 8D 3E 95 A5 B9 D9
9520: 8D 3F 95 A0 01 B1 83 85 F9
9528: B8 C8 B1 83 85 B9 20 C3 67
9530: 94 AD 3E 95 85 B8 AD 3F DB
9538: 95 85 B9 4C B7 00 4C B7 99

```

On Disk Only

If you purchase this issue's *COMPUTE!*'s Apple Applications Disk, you'll find three "Sounding Board" sound files ready to load and enjoy. These sounds are saved as PIANO.S, SQUISH.S, and SPACEY.S. To use any of these files, simply load them in with the sound editor's LOAD option, or install them in your own BASIC programs following the instructions given in the article.

Look for the "On Disk Only" box in all of *Apple Applications*' articles. If a program or article can be enhanced by additional disk files, we'll explain them here and provide them on disk. For more information on ordering *COMPUTE!*'s Apple Applications Disk, see page 29.

aa